

FONCTIONS DE n VARIABLES RÉELLES : DÉFINITION, LIMITE, DÉRIVABILITÉ, DIFFÉRENTIABILITÉ

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Les fonctions de plusieurs variables réelles représentant le plus gros morceau de l'analyse pour économistes et c'est logique car peu de phénomènes économiques ne font entrer en jeu qu'une seule cause. Au passage, rappelons qu'en économie comme dans beaucoup de sciences appliquées, l'écriture $(x_1, x_2, \dots, x_n) \rightarrow y = f(x_1, x_2, \dots, x_n)$ signifie implicitement que la variable y est "expliquée" par les variables "explicatives" x_1, x_2, \dots, x_n . Autrement dit, un ensemble de plusieurs variables exogènes et de paramètres influencent la valeur prise par la variable expliquée ou endogène.

Du point de vue d'un logiciel de calcul formel, il n'y a pas de différence de sens - et même de difficulté - entre les fonctions d'une ou de plusieurs variables réelles. C'est la raison pour laquelle on trouvera dans ce chapitre beaucoup d'instructions similaires à celles du chapitre consacré aux fonctions d'une variable : **unapply**, **limit**, **diff**, **D**, etc.

La première section expose les deux manières de définir une fonction mathématique et les passerelles permettant de passer d'une forme à l'autre.

La section 2 est consacrée à la commande **limit**.

La section 3 porte sur la dérivabilité : calculs de dérivées partielles premières, de dérivées partielles d'ordre n , de gradients et de matrices hessiennes.

Enfin, la différentiabilité est abordée dans la section 4, qui montre comment on obtient le DL(n) d'une fonction puis comment il est possible d'exprimer la différentielle d'une variable expliquée.

▼ DÉFINITION D'UNE FONCTION

Comme pour les fonctions d'une variable réelle, il y a deux manières de définir une fonction de n variables réelles : fonction-procédure ou expression.

▼ Fonction-procédure

On définit une fonction-procédure de plusieurs variables réelles avec l'opérateur **->**, soit la combinaison de touches "tiret"+"supérieur à". A sa gauche, on place les variables explicatives entre parenthèses. A sa droite, on explicite la correspondance. Il est recommandé d'assigner la procédure. Pour connaître la valeur prise par la fonction en un point $X = (x_1, x_2, \dots, x_n)$, on appelle la fonction et on place entre parenthèses les valeurs - nombres ou symboles - désirées.

```
> restart;
F1:=(x,y)->x^2+y^2;#fonction de deux variables
F1(1,1);#calcul de la valeur prise par F1 au point (1;1)
F2:=(x,y,z,t)->(x^3-y^3+z^3+t^3)/(x-y);#fonction de quatre
variables
F2(1,0,alpha,beta);#calcul de la valeur prise par F2 au
point (1;0;alpha;beta)
```

$$F1 := (x, y) \rightarrow x^2 + y^2$$

$$F2 := (x, y, z, t) \rightarrow \frac{x^3 - y^3 + z^3 + t^3}{x - y} \cdot (1 + \alpha^3 + \beta^3)$$

Maple est intraitable sur le domaine de définition de la fonction. Un message d'erreur sanctionne toute demande de calcul infaisable/interdit :

```
> F2(2,2,a,b);
Error, (in F2) numeric exception: division by zero
```

Expression

Bien entendu, rien n'interdit d'utiliser une expression, quoique ce soit bien moins pratique pour calculer la valeur de l'image :

```
> P1:=x^2+y^2;#expression assignée décrivant une fonction de
deux variables
x:=1;y:=1;#attribution de valeurs aux variables
P1;#évaluation de la fonction en (1;1)
x:='x';y:='y';#désassignation de x et de y
eval(P1,{x=1,y=1});#évaluation au moyen de la commande eval
subs({x=1,y=1},P1);#évaluation au moyen de la commande subs
```

$$P1 := x^2 + y^2$$

$$x := 1$$

$$y := 1$$

$$2$$

$$x := x$$

$$y := y$$

$$2$$

$$2$$

Passerelles entre fonction-procédure et expression

On passe d'une expression **Xp** à une fonction-procédure **f** en invoquant la commande **unapply** dont le premier argument est l'expression considérée et les suivants sont les variables explicatives. La syntaxe est donc : **f:=unapply(Xp,x1,x2,...,xn)**. L'output renvoie la fonction-procédure.

```
> restart;
Xp:=(x^2+y^3)/z^(1/3);#écriture d'une expression
f_1:=unapply(Xp,x,y,z);#passage à une fonction de trois
variables
f_2:=unapply(Xp,x,z);#passage à une fonction en x et z
f_3:=unapply(Xp,x,y);#passage à une fonction en x et y
```

$$Xp := \frac{x^2 + y^3}{z^{1/3}}$$

$$f_1 := (x, y, z) \rightarrow \frac{x^2 + y^3}{z^{1/3}}$$

$$f_2 := (x, z) \rightarrow \frac{x^2 + y^3}{z^{1/3}}$$

$$f_3 := (x, y) \rightarrow \frac{x^2 + y^3}{z^{1/3}}$$

Inversement, si **f** est une fonction définie par **(x,y)→f(x,y)**, l'expression correspondante **Xp** est donnée en écrivant simplement **f(x,y)** :

```
> Xp_1:=f_1(x,y,z);#passage à une expression
x:=1;y:=0;z:=2;#attribution de valeurs aux variables
explicatives
Xp_1;#évaluation
x:='x';y:='y';#désassignation de x et de y
```

$$Xp_1 := \frac{1}{2} 2^{2/3}$$

$$x := 1$$

$$y := 0$$

$$z := 2$$

$$\frac{1}{2} 2^{2/3}$$

$$x := x$$

$$y := y$$

▼ LIMITES

La commande **limit** tente de calculer la limite d'une expression en un point. Elle a deux arguments : le premier est l'expression considérée; la seconde donne entre accolades les coordonnées du point considéré sous forme d'égalités. Ainsi, l'instruction **limit(Xp, {x=a, y=b})** demande la limite de l'expression **Xp** quand le point (x,y) tend vers le point (a,b). L'exemple suivant est enfantin :

```
> limit((x^2-y^2)/(x-y), {x=0, y=0});
0
```

Mais cet autre exemple montre que le calcul demandé n'est pas toujours à la portée du logiciel, qui renvoie alors en écho une forme non évaluée. Comme c'est souvent le cas, il faut aider Maple pour obtenir une réponse satisfaisante.

```
> limit(x*y*sin(1/x), {x=0, y=1});
limit(x y sin(1/x), {x=0, y=1})
```

Comme on a toujours $-|x y| \leq x y \sin\left(\frac{1}{x}\right) \leq |x y|$, on fait jouer le théorème des gendarmes :

```
> limit(-(x*y), {x=0, y=1}); limit(x*y, {x=0, y=1});
0
```

Au voisinage du point $(0; 1)$, que x tende vers 0 par valeurs négatives ou par valeurs positives, la limite de la fonction est 0.

▼ DÉRIVABILITÉ

Maple est époustouflant dans le calcul des dérivées partielles, des gradients et des hessiens.

▼ Dérivées partielles des expressions de n variables réelles

Soit une expression Xp contenant plusieurs variables x_1, x_2, \dots, x_n , soit, en notation input, respectivement **Xp, x1, x2, ..., xn**. La commande **diff(Xp, x1)** renvoie la dérivée partielle première de l'expression par rapport à x_1 . Plus généralement, la commande **diff(Xp, xi)** renvoie la dérivée partielle première de l'expression par rapport à $x_i, i = 1, \dots, n$.

```
> restart;
Xp:=A*x^alpha*y^(1-alpha);
Xp1x:=diff(Xp,x);Xply:=diff(Xp,y);#dérivées partielles
d'une Cobb-Douglas
```

$$Xp := A x^\alpha y^{1-\alpha}$$

$$Xp1x := \frac{A x^\alpha \alpha y^{1-\alpha}}{x}$$

$$Xply := \frac{A x^\alpha y^{1-\alpha} (1-\alpha)}{y}$$

La commande **simplify** rend les résultats plus lisibles :

```
> simplify(Xp1x);simplify(Xply);
A x^{-1+\alpha} \alpha y^{1-\alpha}
-A x^\alpha y^{-\alpha} (-1+\alpha)
```

Les dérivées partielles secondes par rapport à x_i puis à x_j sont données par **diff(Xp, xi, xj)** :

```
> r:=simplify(diff(Xp,x,x));#dérivée partielle seconde par
rapport à x deux fois
s:=simplify(diff(Xp,x,y));#dérivée partielle seconde par
rapport à x puis à y
t:=simplify(diff(Xp,y,y));#dérivée partielle seconde par
rapport à y deux fois
```

$$r := A x^{\alpha-2} \alpha y^{1-\alpha} (-1+\alpha)$$

$$s := -A x^{-1+\alpha} \alpha y^{-\alpha} (-1+\alpha)$$

$$t := A x^\alpha y^{-1-\alpha} \alpha (-1+\alpha)$$

On calcule suivant le même principe les dérivées d'ordre supérieur. Il faut noter que Maple suppose toujours que les expressions sont des fonctions continûment dérivables et applique en conséquence le théorème de Schwartz. L'ordre des dérivations importe peu.

Il est alors possible d'utiliser l'écriture raccourcie **diff(Xp, x\$ p , y\$ q)** pour demander la

dérivée partielle d'ordre $(p+q)$ d'une fonction de deux variables, p fois par rapport à x et q fois par rapport à y .

> **diff(exp(a*x*y^2),x\$2,y\$3);**

$$24 a^2 y e^{axy^2} + 96 a^3 y^3 x e^{axy^2} + 60 a^4 y^5 x^2 e^{axy^2} + 8 a^5 y^7 x^3 e^{axy^2}$$

La forme inerte **Diff** n'évalue pas la dérivée partielle; elle renvoie en écho l'écriture mathématique du problème posé :

> **Diff(exp(a*x*y^2),x\$2,y\$3);**

$$\frac{\partial^5}{\partial y^3 \partial x^2} e^{axy^2}$$

Pour connaître la réponse, il faut demander explicitement l'évaluation avec **value** :

> **value(%);**

$$24 a^2 y e^{axy^2} + 96 a^3 y^3 x e^{axy^2} + 60 a^4 y^5 x^2 e^{axy^2} + 8 a^5 y^7 x^3 e^{axy^2}$$

On tirera avantage de cette particularité pour rendre plus lisible sa feuille de travail :

> **Diff(exp(a*x*y^2),x\$2,y\$3):%=value(%);**

$$\frac{\partial^5}{\partial y^3 \partial x^2} e^{axy^2} = 24 a^2 y e^{axy^2} + 96 a^3 y^3 x e^{axy^2} + 60 a^4 y^5 x^2 e^{axy^2} + 8 a^5 y^7 x^3 e^{axy^2}$$

▼ Dérivées partielles des fonctions de n variables réelles

Quand on utilise une fonction-procédure, les dérivées partielles sont données par l'opérateur **D**, qui renvoie une fonction (et non une expression).

La fonction dérivée partielle première de la fonction de n variables f par rapport à la i -ème variable ($0 \leq i \leq n$) est demandée par **D[i](f)** :

> **f:=(x,y,z)->1/(x^2+y^2+z^2)^(1/2);**

D[1](f);D[2](f);D[3](f);

$$f := (x, y, z) \rightarrow \frac{1}{\sqrt{x^2 + y^2 + z^2}}$$

$$(x, y, z) \rightarrow -\frac{x}{(x^2 + y^2 + z^2)^{3/2}}$$

$$(x, y, z) \rightarrow -\frac{y}{(x^2 + y^2 + z^2)^{3/2}}$$

$$(x, y, z) \rightarrow -\frac{z}{(x^2 + y^2 + z^2)^{3/2}}$$

La fonction dérivée partielle seconde de f par rapport à la variable i puis la variable j est demandée par **D[i,j](f)** :

> **D[1,1](f);D[1,2](f);D[1,3](f);D[2,1](f);D[2,2](f);D[2,3](f);**
D[3,1](f);D[3,2](f);D[3,3](f);

$$(x, y, z) \rightarrow \frac{3x^2}{(x^2 + y^2 + z^2)^{5/2}} - \frac{1}{(x^2 + y^2 + z^2)^{3/2}}$$

$$(x, y, z) \rightarrow \frac{3yx}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3zx}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3yx}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3y^2}{(x^2 + y^2 + z^2)^{5/2}} - \frac{1}{(x^2 + y^2 + z^2)^{3/2}}$$

$$(x, y, z) \rightarrow \frac{3zy}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3zx}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3zy}{(x^2 + y^2 + z^2)^{5/2}}$$

$$(x, y, z) \rightarrow \frac{3z^2}{(x^2 + y^2 + z^2)^{5/2}} - \frac{1}{(x^2 + y^2 + z^2)^{3/2}}$$

Pour se repérer, il n'est pas inutile de rappeler avant les résultats l'expression symbolique de la dérivée que l'on obtient en la plaçant entre deux apostrophes, une avant et une après (les apostrophes servent à empêcher l'évaluation immédiate) :

> 'D[1,2,3](f)'=D[1,2,3](f);

$$D_{1,2,3}(f) = \left((x, y, z) \rightarrow -\frac{15zyx}{(x^2 + y^2 + z^2)^{7/2}} \right)$$

L'évaluation d'une dérivée partielle en un point particulier s'obtient en faisant figurer entre parenthèses les coordonnées du point après la fonction dérivée :

> D[1,2,3](f)(1.25,0.75,6.358);#calcul de la dérivée en un point particulier

D[1,2,3](f)(alpha,beta,gamma);#calcul de la dérivée en un point symbolique

-0.0001779362562

$$-\frac{15\gamma\beta\alpha}{(\alpha^2 + \beta^2 + \gamma^2)^{7/2}} \quad (3.2.1)$$

▼ Gradient

Soit f une fonction de plusieurs variables x_1, x_2, \dots, x_n , supposée continûment dérivable. Notons X le point de \mathbb{R}^n de coordonnées x_1, x_2, \dots, x_n .

Si f est écrite en tant qu'expression sous la forme $\mathbf{f}(X)$, son gradient est donné in extenso par : `<diff(f(X),x[1])|diff(f(X),x[2])|...|diff(f(X),x[n])>`. Si f est écrite comme fonction-procédure \mathbf{f} , alors son gradient est donné par la commande `<D[1]`

```
(f)(X),D[2](f)(X),...,D[n](f)(X)>
```

Dans les deux cas, on fait intervenir un vecteur-ligne Maple, qui s'ouvre par < et se ferme par >, et les composantes sont séparées par un trait vertical | (l'écriture d'un vecteur est expliquée dans le chapitre consacré à l'algèbre linéaire).

```
> restart;  
f(X):=1/(x^2+y^2+z^2)^(1/2);#écriture d'une expression de  
trois variables  
<diff(f(X),x)|diff(f(X),y)|diff(f(X),z)>;#gradient d'une  
expression  
whattype(%);
```

$$f(X) := \frac{1}{\sqrt{x^2 + y^2 + z^2}}$$
$$\left[-\frac{x}{(x^2 + y^2 + z^2)^{3/2}} \quad -\frac{y}{(x^2 + y^2 + z^2)^{3/2}} \quad -\frac{z}{(x^2 + y^2 + z^2)^{3/2}} \right]$$

Vector_{row}

```
> f:=(x,y,z)->1/(x^2+y^2+z^2)^(1/2);#écriture d'une fonction  
de trois variables  
<D[1](f)(x,y,z)|D[2](f)(x,y,z)|D[3](f)(x,y,z)>;#gradient  
d'une fonction  
whattype(%);
```

$$f := (x, y, z) \rightarrow \frac{1}{\sqrt{x^2 + y^2 + z^2}}$$
$$\left[-\frac{x}{(x^2 + y^2 + z^2)^{3/2}} \quad -\frac{y}{(x^2 + y^2 + z^2)^{3/2}} \quad -\frac{z}{(x^2 + y^2 + z^2)^{3/2}} \right]$$

Vector_{row}

Au point $X_0 = (x_0, y_0, z_0)$, on a :

```
> <D[1](f)(x[0],y[0],z[0])|D[2](f)(x[0],y[0],z[0])|D[3](f)(x  
[0],y[0],z[0])>;
```

$$\left[-\frac{x_0}{(x_0^2 + y_0^2 + z_0^2)^{3/2}} \quad -\frac{y_0}{(x_0^2 + y_0^2 + z_0^2)^{3/2}} \quad -\frac{z_0}{(x_0^2 + y_0^2 + z_0^2)^{3/2}} \right]$$

Mais d'autres manières d'obtenir un gradient existent. Ainsi, le paquetage **VectorCalculus** propose la commande **Gradient** qui, comme son nom l'indique sans ambiguïté, renvoie le gradient d'une fonction. On donne en premier argument l'expression et en second argument la liste des variables indépendantes. Le résultat est un vecteur dont les composantes dans la base canonique sont les dérivées partielles premières de la fonction.

```
> with(VectorCalculus):gradf:=Gradient(f(x,y,z),[x,y,z]);
```

$$\text{gradf} := -\frac{x}{(x^2 + y^2 + z^2)^{3/2}} \bar{e}_x - \frac{y}{(x^2 + y^2 + z^2)^{3/2}} \bar{e}_y - \frac{z}{(x^2 + y^2 + z^2)^{3/2}} \bar{e}_z$$

Ce résultat est peu exploitable en économie. On récupère le vecteur-ligne sous forme de liste ou de vecteur par la commande **convert** :

```
> convert(gradf,list);
whattype(%);
convert(gradf,Vector,orientation=row);
whattype(%);
```

$$\left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

list

$$\left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

Vector_{row}

Une autre façon de procéder est de recourir au paquetage **linalg** qui contient la commande **grad(Xp,[variables])** où **Xp** est une expression dépendant de plusieurs variables et **[variables]** spécifie dans une liste les variables indépendantes. Le résultat est un tableau à une dimension (array) qu'on peut transformer à sa guise en liste ou vecteur par **convert**.

```
> with(linalg):DfX:=grad(f(x,y,z),[x,y,z]);whattype(%);
convert(DfX,list);whattype(%);
convert(DfX,Vector);whattype(%);
```

$$DfX := \left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

array

$$\left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

list

$$\left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

Vector_{row}

▼ Matrice hessienne

Supposons à présent que la fonction f soit de classe C^2 dans \mathbb{R}^n et qu'on ait besoin de sa matrice hessienne.

Si f est manipulée comme expression sous la forme **f(X)**, sa hessienne est écrite in extenso par les deux commandes équivalentes :

- **Matrix(n,n,[diff(f(X),x1,x1),diff(f(X),x1,x2),...,diff(f(X),x1,xn)],...,[diff(f(X),xn,x1),diff(f(X),xn,x2),...,diff(f(X),xn,xn)])]**)
- **<<diff(f(X),x1,x1)|diff(f(X),x1,x2)|...|diff(f(X),x1,xn)>,...**

`., <diff(f(X), xn, x1) | diff(f(X), xn, x2) | ... | diff(f(X), xn, xn)>>`

Si f est écrite comme fonction-procédure f , alors on pose :

- `Matrix(n, n, [[D[1,1](f)(X), D[1,2](f)(X), ..., D[1,n](f)(X)], ..., [D[n,1](f)(X), D[n,2](f)(X), ..., D[n,n](f)(X)]])`
- `<<D[1,1](f)(X) | D[1,2](f)(X) | ... | D[1,n](f)(X)>>, ..., <D[n,1](f)(X) | D[n,2](f)(X) | ... | D[n,n](f)(X)>>`

Dans les deux cas, on fait intervenir une matrice Maple, dont on verra la syntaxe dans le chapitre consacré à l'algèbre linéaire. A ce niveau, le plus simple est de déclarer une matrice par `Matrix()`, puis de donner ses dimensions dans les deux premiers arguments (ici : n, n), enfin de construire une liste de n listes dans lesquelles figurent les dérivées partielles secondes en ligne ... et dans le bon ordre.

`> restart;`

`fonc(X):=1/(x^2+y^2+z^2)^(1/2):#écriture d'une expression`
`hessex:=Matrix(3,3,[[diff(fonc(X),x,x),diff(fonc(X),x,y),`
`diff(fonc(X),x,z)], [diff(fonc(X),y,x),diff(fonc(X),y,y),`
`diff(fonc(X),y,z)], [diff(fonc(X),z,x),diff(fonc(X),z,y),`
`diff(fonc(X),z,z)]]);#hessienne de l'expression en un point`
`quelconque`

$$\text{hessex} := \left[\left[\frac{3x^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3xz}{(x^2+y^2+z^2)^{5/2}} \right], \left[\frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3y^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}} \right], \left[\frac{3xz}{(x^2+y^2+z^2)^{5/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}}, \frac{3z^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}} \right] \right]$$

`> f:=(x,y,z)->1/(x^2+y^2+z^2)^(1/2):#écriture d'une fonction-procédure`

`hessf:=Matrix(3,3,[[D[1,1](f)(x,y,z),D[1,2](f)(x,y,z),D[1,3](f)(x,y,z)], [D[2,1](f)(x,y,z),D[2,2](f)(x,y,z),D[2,3](f)(x,y,z)], [D[3,1](f)(x,y,z),D[3,2](f)(x,y,z),D[3,3](f)(x,y,z)]]);#hessienne de la fonction en un point X quelconque`

$$\text{hessf} := \left[\left[\frac{3x^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3xz}{(x^2+y^2+z^2)^{5/2}} \right], \left[\frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3y^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}} \right], \left[\frac{3xz}{(x^2+y^2+z^2)^{5/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}}, \frac{3z^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}} \right] \right]$$

$$\left[\frac{3xz}{(x^2+y^2+z^2)^{5/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}}, \frac{3z^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}} \right]$$

L'avantage des fonctions-procédures est d'exécuter très facilement des évaluations. Ainsi, au point $X_0 = (x_0, y_0, z_0)$, on a :

```
> hessfX0:=Matrix(3,3,['Diff(f(X[0]),x,x)', 'Diff(f(X[0]),x,
y)', 'Diff(f(X[0]),x,z)'], ['Diff(f(X[0]),y,x)', 'Diff(f(X[0]
),y,y)', 'Diff(f(X[0]),y,z)'], ['Diff(f(X[0]),z,x)', 'Diff(f
(X[0]),z,y)', 'Diff(f(X[0]),z,z)'])=Matrix(3,3,[[D[1,1](f)
(x[0],y[0],z[0]),D[1,2](f)(x[0],y[0],z[0]),D[1,3](f)(x[0],y
[0],z[0])],[D[2,1](f)(x[0],y[0],z[0]),D[2,2](f)(x[0],y[0],z
[0]),D[2,3](f)(x[0],y[0],z[0])],[D[3,1](f)(x[0],y[0],z[0]
),D[3,2](f)(x[0],y[0],z[0]),D[3,3](f)(x[0],y[0],z[0])]]);
```

$$\text{hessfX0} := \begin{bmatrix} \frac{\partial^2}{\partial x^2} f(X_0) & \frac{\partial^2}{\partial y \partial x} f(X_0) & \frac{\partial^2}{\partial z \partial x} f(X_0) \\ \frac{\partial^2}{\partial x \partial y} f(X_0) & \frac{\partial^2}{\partial y^2} f(X_0) & \frac{\partial^2}{\partial z \partial y} f(X_0) \\ \frac{\partial^2}{\partial x \partial z} f(X_0) & \frac{\partial^2}{\partial y \partial z} f(X_0) & \frac{\partial^2}{\partial z^2} f(X_0) \end{bmatrix} = \left[\left[\frac{3x_0^2}{(x_0^2+y_0^2+z_0^2)^{5/2}} \right. \right. \quad (3.4.1)$$

$$\left. \left. - \frac{1}{(x_0^2+y_0^2+z_0^2)^{3/2}}, \frac{3y_0x_0}{(x_0^2+y_0^2+z_0^2)^{5/2}}, \frac{3z_0x_0}{(x_0^2+y_0^2+z_0^2)^{5/2}} \right], \right.$$

$$\left[\frac{3y_0x_0}{(x_0^2+y_0^2+z_0^2)^{5/2}}, \frac{3y_0^2}{(x_0^2+y_0^2+z_0^2)^{5/2}} - \frac{1}{(x_0^2+y_0^2+z_0^2)^{3/2}}, \frac{3z_0y_0}{(x_0^2+y_0^2+z_0^2)^{5/2}} \right],$$

$$\left[\frac{3z_0x_0}{(x_0^2+y_0^2+z_0^2)^{5/2}}, \frac{3z_0y_0}{(x_0^2+y_0^2+z_0^2)^{5/2}}, \frac{3z_0^2}{(x_0^2+y_0^2+z_0^2)^{5/2}} - \frac{1}{(x_0^2+y_0^2+z_0^2)^{3/2}} \right]$$

Une méthode plus rapide consiste à invoquer la commande **Hessian** du paquetage **VectorCalculus**. Son premier argument est une expression représentant la fonction de plusieurs variables étudiée et le second est une liste formée des variables. Le résultat est une matrice carrée dont les éléments sont des expressions. Par conséquent, pour calculer la matrice hessienne en un point, il faut préalablement spécifier qu'elle dépend des variables par **unapply**

```
> with(VectorCalculus):#chargement du paquetage
Hessian(f(x,y,z),[x,y,z]);#requête pour une expression
quelconque
whattype(%);
H:=unapply(%, [x,y,z]):#transformation des composantes de
la hessienne en fonctions-procédures de plusieurs variables
H(0,1,2);#calcul de la hessienne en un point
```

$$\left[\left[\frac{3x^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3xz}{(x^2+y^2+z^2)^{5/2}} \right], \right. \\ \left[\frac{3xy}{(x^2+y^2+z^2)^{5/2}}, \frac{3y^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}} \right], \\ \left. \left[\frac{3xz}{(x^2+y^2+z^2)^{5/2}}, \frac{3yz}{(x^2+y^2+z^2)^{5/2}}, \frac{3z^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}} \right] \right]$$

Matrix

$$\begin{bmatrix} -\frac{1}{25}\sqrt{5} & 0 & 0 \\ 0 & -\frac{2}{125}\sqrt{5} & \frac{6}{125}\sqrt{5} \\ 0 & \frac{6}{125}\sqrt{5} & \frac{7}{125}\sqrt{5} \end{bmatrix}$$

Alternativement, on peut également utiliser la commande `hessian(xp,[variables])` du module `linalg`. Sa syntaxe est celle de `Hessian` mais l'output est du type `array` et non `Matrix`. Si nécessaire, on passe du tableau à une matrice par `convert`.

```
> restart;
with(linalg):#chargement du paquetage
hessian(f(x,y),[x,y]);#requête pour une expression
quelconque
whattype(%);
convert(%,Matrix);#passage d'un tableau à une matrice
whattype(%);
hessian(1/(x^2+y^2+z^2)^(1/2),[x,y,z]);#reprise de
l'exemple précédent
```

$$\begin{bmatrix} \frac{\partial^2}{\partial x^2} f(x,y) & \frac{\partial^2}{\partial y \partial x} f(x,y) \\ \frac{\partial^2}{\partial y \partial x} f(x,y) & \frac{\partial^2}{\partial y^2} f(x,y) \end{bmatrix}$$

array

$$\begin{bmatrix} \frac{\partial^2}{\partial x^2} f(x,y) & \frac{\partial^2}{\partial y \partial x} f(x,y) \\ \frac{\partial^2}{\partial y \partial x} f(x,y) & \frac{\partial^2}{\partial y^2} f(x,y) \end{bmatrix}$$

Matrix

$$\left[\left[\frac{3x^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3yx}{(x^2+y^2+z^2)^{5/2}}, \frac{3zx}{(x^2+y^2+z^2)^{5/2}} \right], \right.$$

$$\left[\frac{3yx}{(x^2+y^2+z^2)^{5/2}}, \frac{3y^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}}, \frac{3zy}{(x^2+y^2+z^2)^{5/2}} \right]$$

$$\left[\frac{3zx}{(x^2+y^2+z^2)^{5/2}}, \frac{3zy}{(x^2+y^2+z^2)^{5/2}}, \frac{3z^2}{(x^2+y^2+z^2)^{5/2}} - \frac{1}{(x^2+y^2+z^2)^{3/2}} \right]$$

▼ DÉVELOPPEMENTS LIMITÉS et DIFFÉRENTIELLES

Les économistes raffolent des différentielles de fonctions de plusieurs variables réelles. L'ennui est que Maple offre peu de possibilités dans ce domaine. Il est cependant possible d'obtenir aisément la partie régulière d'un développement limité à l'ordre $(n-1)$ par la commande **mtaylor(Xp, [variables], n)** dans laquelle **Xp** représente une expression faisant intervenir plusieurs variables **x1,x2,...,xn**; **[variables]** est la liste des variables - auquel cas on suppose que le développement est fait à l'origine, ou une liste d'égalités - auquel cas on précise les coordonnées du point au voisinage duquel le développement est effectué, et **n** est l'ordre du développement augmenté d'une unité.

```
> restart;
mtaylor(f(x,y),[x,y],2);#partie principale d'un DL(1)au
voisinage de (0,0)
mtaylor(f(x,y),[x,y],3);#partie principale d'un DL(2)au
voisinage de (0,0)
```

$$f(0,0) + D_1(f)(0,0)x + D_2(f)(0,0)y$$

$$f(0,0) + D_1(f)(0,0)x + D_2(f)(0,0)y + \frac{1}{2} D_{1,1}(f)(0,0)x^2 + x D_{1,2}(f)(0,0)y$$

$$+ \frac{1}{2} D_{2,2}(f)(0,0)y^2$$

```
> restart;
mtaylor(f(x,y),[x=x0,y=y0],2);#partie principale d'un DL(1)au
voisinage de (x0,y0)
mtaylor(f(x,y),[x=x0,y=y0],3);#partie principale d'un DL(2)au
voisinage de (x0,y0)
```

$$f(x_0, y_0) + D_1(f)(x_0, y_0)(x - x_0) + D_2(f)(x_0, y_0)(y - y_0)$$

$$f(x_0, y_0) + D_1(f)(x_0, y_0)(x - x_0) + D_2(f)(x_0, y_0)(y - y_0) + \frac{1}{2} D_{1,1}(f)(x_0, y_0)(x$$

$$- x_0)^2 + (x - x_0) D_{1,2}(f)(x_0, y_0)(y - y_0) + \frac{1}{2} D_{2,2}(f)(x_0, y_0)(y - y_0)^2$$

```
> restart;
xpl:=(x,y)->cos(x+y)+sin(x+y);#création d'une fonction-
procédure
mtaylor(xpl(x,y),[x,y],2);
mtaylor(xpl(x,y),[x,y],3);
```

$$xpl := (x, y) \rightarrow \cos(x + y) + \sin(x + y)$$

$$1 + x + y$$

$$1 + x + y - \frac{1}{2} x^2 - yx - \frac{1}{2} y^2$$

```
> restart;
xp2:=(x,y)->x^alpha*y^(1-alpha);#exemple d'une Cobb-Douglas
DL1(x0,y0):=simplify(mtaylor(xp2(x,y),[x=x0,y=y0],2));
DL2(x0,y0):=simplify(mtaylor(xp2(x,y),[x=x0,y=y0],3));
```

$$xp2 := (x, y) \rightarrow x^\alpha y^{1-\alpha}$$

$$DL1(x0, y0) := x0^\alpha y0^{-\alpha} y - x0^\alpha y0^{-\alpha} \alpha y + x0^{-1+\alpha} \alpha y0^{1-\alpha} x$$

$$DL2(x0, y0) := x0^\alpha y0^{-\alpha} y - x0^\alpha y0^{-\alpha} \alpha y + x0^{-1+\alpha} \alpha y0^{1-\alpha} x - \frac{1}{2} x0^{\alpha-2} \alpha y0^{1-\alpha} x^2$$

$$- \frac{1}{2} x0^\alpha y0^{-1-\alpha} \alpha y^2 + \frac{1}{2} x0^{\alpha-2} \alpha^2 y0^{1-\alpha} x^2 + \frac{1}{2} x0^\alpha y0^{-1-\alpha} \alpha^2 y^2$$

$$+ x0^{-1+\alpha} \alpha y0^{-\alpha} yx - x0^{-1+\alpha} \alpha^2 y0^{-\alpha} xy$$

Reste que la question des différentielles, telles qu'elles sont pratiquées par les économistes (et les physiciens), reste posée puisque Maple ne prend pas en charge cette notion. Il est cependant possible de les introduire dans une feuille de travail.

Une première méthode, qu'on peut qualifier de "bourrin", consiste à appeler les paquets **linalg** et **LinearAlgebra**, utiliser avec la première la commande **grad** (non sans avoir converti le résultat du type **Array** en type **Vector**), puis avec la seconde la commande **DotProduct** (avec l'option **conjugate=false**) :

```
> restart;with(linalg):with(LinearAlgebra):gr:=convert(grad(f
(x1,x2),[x1,x2]),Vector);v:=<dx1,dx2>;dy=DotProduct(gr,v,
conjugate=false);
```

$$gr := \begin{bmatrix} \frac{\partial}{\partial x1} f(x1, x2) & \frac{\partial}{\partial x2} f(x1, x2) \end{bmatrix}$$

$$v := \begin{bmatrix} dx1 \\ dx2 \end{bmatrix}$$

$$dy = dx1 \left(\frac{\partial}{\partial x1} f(x1, x2) \right) + dx2 \left(\frac{\partial}{\partial x2} f(x1, x2) \right)$$

Une deuxième méthode, très élégante, basée sur une utilisation astucieuse de la commande **add**, nous a été transmise par Frédéric Dupont (Université de Brest). On crée une fonction Maple dont les arguments sont d'une part une fonction anonyme **f** et d'autre part un ensemble anonyme **V** et l'image la somme des différentielles des variables appartenant à l'ensemble pondérées par les dérivées partielles. La différentielle d'une variable est obtenue par concaténation de la lettre d et de la variable. A un tel niveau de généralité, la technique fonctionne pour une fonction de une variable ou de plusieurs variables.

```
> restart;
Dt:=(f,V)->add(diff(f,v)*cat(d,v),v=V);#fonction-procédure
générant la différentielle totale de la fonction f
```

$$Dt := (f, V) \rightarrow add \left(\left(\frac{\partial}{\partial v} f \right) cat(d, v), v = V \right)$$

En particulierisant la fonction-expression \mathbf{f} et l'ensemble anonyme \mathbf{V} , on obtient la différentielle cherchée :

> `dy:=simplify(Dt(x^n,{x}));#exemple d'une fonction d'une variable`

$$dy := x^{n-1} n dx \quad (4.1)$$

> `dy:=simplify(Dt(x^n,{x,n}));#exemple d'une fonction de deux variables`

$$dy := x^n \ln(x) dn + x^{n-1} n dx$$

> `dY:=simplify(Dt(A*K^alpha*L^(1-alpha),{L,K}),power);
#différentielle de la fonction Cobb-Douglas`

$$dY := A K^{-1+\alpha} \alpha L^{1-\alpha} dK + A K^\alpha L^{-\alpha} (1-\alpha) dL$$

Exercices

Exercice M1

Soit la fonction $f(x, y) = \frac{x^\alpha y}{x^2 + y^2}$.

1. Calculer les dérivées partielles de f .
2. Calculer les élasticités de f par rapport à x et à y .

Exercice M2

Calculer une valeur approchée de $x^\alpha y^{1-\alpha}$ au point $(x, y) = (10.1, 9.95)$.

Exercice M3

Calculer la différentielle de $f(x, y, z) = (x^2 + y) e^z$ en $M = (a, b, c)$.

Exercice M4

Soit $f : \mathbb{R}^2 \rightarrow \mathbb{R}$, une fonction de classe C^3 et soit $M = (a, b)$. On considère la fonction d'une variable réelle $g : t \rightarrow g(t) = f(M + tH)$ où $H = (u, v)$.

Calculer $g'(t)$ puis $g''(t)$ et $g'''(t)$.

Application : $f(x, y) = x^2 e^y$, $M = (15; 7)$, $H = (1, 1)$.

Exercice E1

Dans le Capital de Karl Marx, le taux de profit, tp , est égal au rapport de la plus-value absolue, pl , et du capital-argent engagé par les capitalistes dans la production, $c + v$, où c exprime la valeur du capital constant et v le capital variable (qui rémunère les travailleurs).

1. On appelle plus-value relative et on note p le rapport de la plus-value absolue et du capital variable. On appelle composition organique du capital et on note k le rapport du capital constant au capital variable. Montrer que le taux de profit est une fonction de p et k .
2. Calculer la différentielle du taux de profit en fonction des différentielles de p et k .

▼ Exercice E2

Une fonction de production dite CES (Constant Elasticity of Substitution) s'écrit :

$$q(x, y) = A \left[\alpha x^{-\rho} + (1 - \alpha) y^{-\rho} \right]^{-\frac{1}{\rho}}$$

où x et y représentent deux facteurs de production, A est un facteur d'échelle (réel strictement positif), α est un réel strictement compris entre 0 et 1 et ρ est un paramètre réel.

1. Montrer qu'une fonction Cobb-Douglas est une fonction CES pour laquelle $\rho = 0$.

2. Calculer le taux de substitution technique $\frac{\frac{\partial q}{\partial x}(x, y)}{\frac{\partial q}{\partial y}(x, y)}$ puis montrer qu'il est décroissant par rapport à x si $\rho > -1$.

3. Calculer la matrice hessienne d'une CES par rapport aux deux facteurs de production puis son déterminant.